

NAG Library Routine Document

F04CAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F04CAF computes the solution to a complex system of linear equations $AX = B$, where A is an n by n matrix and X and B are n by r matrices. An estimate of the condition number of A and an error bound for the computed solution are also returned.

2 Specification

```
SUBROUTINE F04CAF (N, NRHS, A, LDA, IPIV, B, LDB, RCOND, ERBND, IFAIL)
INTEGER          N, NRHS, LDA, IPIV(N), LDB, IFAIL
REAL (KIND=nag_wp) RCOND, ERBND
COMPLEX (KIND=nag_wp) A(LDA,*), B(LDB,*)
```

3 Description

The LU decomposition with partial pivoting and row interchanges is used to factor A as $A = PLU$, where P is a permutation matrix, L is unit lower triangular, and U is upper triangular. The factored form of A is then used to solve the system of equations $AX = B$.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia <http://www.netlib.org/lapack/lug>

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

5 Arguments

- 1: N – INTEGER *Input*
On entry: the number of linear equations n , i.e., the order of the matrix A .
Constraint: $N \geq 0$.
- 2: NRHS – INTEGER *Input*
On entry: the number of right-hand sides r , i.e., the number of columns of the matrix B .
Constraint: $NRHS \geq 0$.
- 3: A(LDA,*) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array A must be at least $\max(1, N)$.
On entry: the n by n coefficient matrix A .
On exit: if $IFAIL \geq 0$, the factors L and U from the factorization $A = PLU$. The unit diagonal elements of L are not stored.

- 4: LDA – INTEGER *Input*
On entry: the first dimension of the array A as declared in the (sub)program from which F04CAF is called.
Constraint: $LDA \geq \max(1, N)$.
- 5: IPIV(N) – INTEGER array *Output*
On exit: if IFAIL ≥ 0 , the pivot indices that define the permutation matrix P ; at the i th step row i of the matrix was interchanged with row IPIV(i). IPIV(i) = i indicates a row interchange was not required.
- 6: B(LDB, *) – COMPLEX (KIND=nag_wp) array *Input/Output*
Note: the second dimension of the array B must be at least $\max(1, NRHS)$.
On entry: the n by r matrix of right-hand sides B .
On exit: if IFAIL = 0 or $N + 1$, the n by r solution matrix X .
- 7: LDB – INTEGER *Input*
On entry: the first dimension of the array B as declared in the (sub)program from which F04CAF is called.
Constraint: $LDB \geq \max(1, N)$.
- 8: RCOND – REAL (KIND=nag_wp) *Output*
On exit: if no constraints are violated, an estimate of the reciprocal of the condition number of the matrix A , computed as $RCOND = 1 / (\|A\|_1 \|A^{-1}\|_1)$.
- 9: ERRBND – REAL (KIND=nag_wp) *Output*
On exit: if IFAIL = 0 or $N + 1$, an estimate of the forward error bound for a computed solution \hat{x} , such that $\|\hat{x} - x\|_1 / \|x\|_1 \leq ERRBND$, where \hat{x} is a column of the computed solution returned in the array B and x is the corresponding column of the exact solution X . If RCOND is less than **machine precision**, then ERRBND is returned as unity.
- 10: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL > 0 and IFAIL $\leq N$

Diagonal element (*value*) of the upper triangular factor is zero. The factorization has been completed, but the solution could not be computed.

IFAIL = N + 1

A solution has been computed, but RCOND is less than *machine precision* so that the matrix A is numerically singular.

IFAIL = -1

On entry, N = $\langle value \rangle$.
Constraint: $N \geq 0$.

IFAIL = -2

On entry, NRHS = $\langle value \rangle$.
Constraint: NRHS ≥ 0 .

IFAIL = -4

On entry, LDA = $\langle value \rangle$ and N = $\langle value \rangle$.
Constraint: LDA $\geq \max(1, N)$.

IFAIL = -7

On entry, LDB = $\langle value \rangle$ and N = $\langle value \rangle$.
Constraint: LDB $\geq \max(1, N)$.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

The complex allocatable memory required is $2 \times N$, and the real allocatable memory required is $2 \times N$. In this case the factorization and the solution X have been computed, but RCOND and ERRBND have not been computed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The computed solution for a single right-hand side, \hat{x} , satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and ϵ is the *machine precision*. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$, the condition number of A with respect to the solution of the linear equations. F04CAF uses the approximation $\|E\|_1 = \epsilon \|A\|_1$ to estimate ERRBND. See Section 4.4 of Anderson *et al.* (1999) for further details.

8 Parallelism and Performance

F04CAF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F04CAF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations required to solve the equations $AX = B$ is proportional to $(\frac{2}{3}n^3 + n^2r)$. The condition number estimation typically requires between four and five solves and never more than eleven solves, following the factorization.

In practice the condition number estimator is very reliable, but it can underestimate the true condition number; see Section 15.3 of Higham (2002) for further details.

The real analogue of F04CAF is F04BAF.

10 Example

This example solves the equations

$$AX = B,$$

where

$$A = \begin{pmatrix} -1.34 + 2.55i & 0.28 + 3.17i & -6.39 - 2.20i & 0.72 - 0.92i \\ -0.17 - 1.41i & 3.31 - 0.15i & -0.15 + 1.34i & 1.29 + 1.38i \\ -3.29 - 2.39i & -1.91 + 4.42i & -0.14 - 1.35i & 1.72 + 1.35i \\ 2.41 + 0.39i & -0.56 + 1.47i & -0.83 - 0.69i & -1.96 + 0.67i \end{pmatrix}$$

and

$$B = \begin{pmatrix} 26.26 + 51.78i & 31.32 - 6.70i \\ 6.43 - 8.68i & 15.86 - 1.42i \\ -5.75 + 25.31i & -2.15 + 30.19i \\ 1.16 + 2.57i & -2.56 + 7.55i \end{pmatrix}.$$

An estimate of the condition number of A and an approximate error bound for the computed solutions are also printed.

10.1 Program Text

```

Program f04cafe

!      F04CAF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
      Use nag_library, Only: f04caf, nag_wp, x04dbf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: errbnd, rcond
      Integer                     :: i, ierr, ifail, lda, ldb, n, nrhs
!      .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:,,:), b(:,,:)

```

```

Integer, Allocatable          :: ipiv(:)
Character (1)                 :: clabs(1), rlabs(1)
! .. Executable Statements ..
Write (nout,*) 'F04CAF Example Program Results'
Write (nout,*)
Flush (nout)
! Skip heading in data file
Read (nin,*)
Read (nin,*) n, nrhs
lda = n
ldb = n
Allocate (a(lda,n),b(ldb,nrhs),ipiv(n))
! Read A and B from data file

Read (nin,*)(a(i,1:n),i=1,n)
Read (nin,*)(b(i,1:nrhs),i=1,n)

! Solve the equations AX = B for X

! ifail: behaviour on error exit
!       =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 1
Call f04caf(n,nrhs,a,lda,ipiv,b,ldb,rcond,errbnd,ifail)

If (ifail==0) Then
!   Print solution, estimate of condition number and approximate
!   error bound

ierr = 0
Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed',' ','Solution', &
'Integer',rlabs,'Integer',clabs,80,0,ierr)

Write (nout,*)
Write (nout,*) 'Estimate of condition number'
Write (nout,99999) 1.0E0_nag_wp/rcond
Write (nout,*)
Write (nout,*) 'Estimate of error bound for computed solutions'
Write (nout,99999) errbnd
Else If (ifail==n+1) Then
!   Matrix is numerically singular. Print estimate of
!   reciprocal of condition number and solution
Write (nout,*)
Write (nout,*) 'Estimate of reciprocal of condition number'
Write (nout,99999) rcond
Write (nout,*)
Flush (nout)

ierr = 0
Call x04dbf('General',' ',n,nrhs,b,ldb,'Bracketed',' ','Solution', &
'Integer',rlabs,'Integer',clabs,80,0,ierr)

Else If (ifail>0 .And. ifail<=n) Then
!   The upper triangular matrix U is exactly singular. Print
!   details of factorization
Write (nout,*)
Flush (nout)

ierr = 0
Call x04dbf('General',' ',n,n,a,lda,'Bracketed','F7.4', &
'Details of factorization','Integer',rlabs,'Integer',clabs,80,0, &
ierr)

!   Print pivot indices
Write (nout,*)
Write (nout,*) 'Pivot indices'
Write (nout,99998) ipiv(1:n)
Else
Write (nout,99997) ifail
End If

```

```

99999 Format (8X,1P,E9.1)
99998 Format ((1X,7I11))
99997 Format (1X,' ** F04CAF returned with IFAIL = ',I5)
      End Program f04cafe

```

10.2 Program Data

F04CAF Example Program Data

```

      4          2          : n, nrhs

(-1.34,  2.55) ( 0.28,  3.17) (-6.39, -2.20) ( 0.72, -0.92)
(-0.17, -1.41) ( 3.31, -0.15) (-0.15,  1.34) ( 1.29,  1.38)
(-3.29, -2.39) (-1.91,  4.42) (-0.14, -1.35) ( 1.72,  1.35)
( 2.41,  0.39) (-0.56,  1.47) (-0.83, -0.69) (-1.96,  0.67) : matrix A

(26.26, 51.78) (31.32, -6.70)
( 6.43, -8.68) (15.86, -1.42)
(-5.75, 25.31) (-2.15, 30.19)
( 1.16,  2.57) (-2.56,  7.55) : matrix B

```

10.3 Program Results

F04CAF Example Program Results

Solution

```

          1          2
1 (  1.0000,  1.0000) ( -1.0000, -2.0000)
2 (  2.0000, -3.0000) (  5.0000,  1.0000)
3 ( -4.0000, -5.0000) ( -3.0000,  4.0000)
4 (  0.0000,  6.0000) (  2.0000, -3.0000)

```

Estimate of condition number
1.5E+02

Estimate of error bound for computed solutions
1.7E-14
